

Stream Fusion

Brian Huffman

March 17, 2025

Abstract

Stream Fusion [1] is a system for removing intermediate list structures from Haskell programs; it consists of a Haskell library along with several compiler rewrite rules. (The library is available online at <http://www.cse.unsw.edu.au/~dons/streams.html>.)

These theories contain a formalization of much of the Stream Fusion library in HOLCF. Lazy list and stream types are defined, along with coercions between the two types, as well as an equivalence relation for streams that generate the same list. List and stream versions of `map`, `filter`, `foldr`, `enumFromTo`, `append`, `zipWith`, and `concatMap` are defined, and the stream versions are shown to respect stream equivalence.

Contents

1	Lazy Lists	2
2	Stream Iterators	3
2.1	Type definitions for streams	4
2.2	Converting from streams to lists	4
2.3	Converting from lists to streams	5
2.4	Bisimilarity relation on streams	6
3	Stream Fusion	6
3.1	Type constructors for state types	6
3.2	Map function	7
3.3	Filter function	8
3.4	Foldr function	9
3.5	EnumFromTo function	10
3.6	Append function	11
3.7	ZipWith function	14
3.8	ConcatMap function	17
3.9	Examples	20

1 Lazy Lists

```

theory LazyList
imports HOLCF HOLCF-Library.Int-Discrete
begin

domain 'a LList = LNil | LCons ('a) ('a LList)

fixrec
mapL :: ('a → 'b) → 'a LList → 'b LList
where
mapL.f.LNil = LNil
| mapL.f.(LCons.x_xs) = LCons.(f.x).(mapL.f_xs)

lemma mapL-strict [simp]: mapL.f.⊥ = ⊥
by fixrec-simp

fixrec
filterL :: ('a → tr) → 'a LList → 'a LList
where
filterL.p.LNil = LNil
| filterL.p.(LCons.x_xs) =
(If p.x then LCons.x.(filterL.p_xs) else filterL.p_xs)

lemma filterL-strict [simp]: filterL.p.⊥ = ⊥
by fixrec-simp

fixrec
foldrL :: ('a → 'b → 'b) → 'b → 'a LList → 'b
where
foldrL.f.z.LNil = z
| foldrL.f.z.(LCons.x_xs) = f.x.(foldrL.f.z_xs)

lemma foldrL-strict [simp]: foldrL.f.z.⊥ = ⊥
by fixrec-simp

fixrec
enumFromToL :: int_⊥ → int_⊥ → (int_⊥) LList
where
enumFromToL.(up.x).(up.y) =
(if x ≤ y then LCons.(up.x).(enumFromToL.(up.(x+1)).(up.y)) else LNil)

lemma enumFromToL-simps' [simp]:
x ≤ y ==>
enumFromToL.(up.x).(up.y) = LCons.(up.x).(enumFromToL.(up.(x+1)).(up.y))
¬ x ≤ y ==> enumFromToL.(up.x).(up.y) = LNil
by simp-all

declare enumFromToL.simps [simp del]

```

```

lemma enumFromToL-strict [simp]:
  enumFromToL· $\perp$ ·y =  $\perp$ 
  enumFromToL·x· $\perp$  =  $\perp$ 
apply (subst enumFromToL.unfold, simp)
apply (induct x)
apply (subst enumFromToL.unfold, simp) +
done

fixrec
  appendL :: 'a LList  $\rightarrow$  'a LList  $\rightarrow$  'a LList
where
  appendL·LNil·ys = ys
  | appendL·(LCons·x·xs)·ys = LCons·x·(appendL·xs·ys)

lemma appendL-strict [simp]: appendL· $\perp$ ·ys =  $\perp$ 
by fixrec-simp

lemma appendL-LNil-right: appendL·xs·LNil = xs
by (induct xs) simp-all

fixrec
  zipWithL :: ('a  $\rightarrow$  'b  $\rightarrow$  'c)  $\rightarrow$  'a LList  $\rightarrow$  'b LList  $\rightarrow$  'c LList
where
  zipWithL·f·LNil·ys = LNil
  | zipWithL·f·(LCons·x·xs)·LNil = LNil
  | zipWithL·f·(LCons·x·xs)·(LCons·y·ys) = LCons·(f·x·y)·(zipWithL·f·xs·ys)

lemma zipWithL-strict [simp]:
  zipWithL·f· $\perp$ ·ys =  $\perp$ 
  zipWithL·f·(LCons·x·xs)· $\perp$  =  $\perp$ 
by fixrec-simp+

fixrec
  concatMapL :: ('a  $\rightarrow$  'b LList)  $\rightarrow$  'a LList  $\rightarrow$  'b LList
where
  concatMapL·f·LNil = LNil
  | concatMapL·f·(LCons·x·xs) = appendL·(f·x)·(concatMapL·f·xs)

lemma concatMapL-strict [simp]: concatMapL·f· $\perp$  =  $\perp$ 
by fixrec-simp

end

```

2 Stream Iterators

```

theory Stream
imports LazyList
begin

```

2.1 Type definitions for streams

Note that everything is strict in the state type.

```
domain ('a,'s) Step = Done | Skip 's | Yield (lazy 'a) 's

type-synonym ('a, 's) Stepper = 's → ('a, 's) Step

domain ('a,'s) Stream = Stream (lazy ('a, 's) Stepper) 's
```

2.2 Converting from streams to lists

```
fixrec
  unfold :: ('a, 's) Stepper -> ('s -> 'a LList)
where
  unfold·h·⊥ = ⊥
  | s ≠ ⊥ =>
    unfold·h·s =
    (case h·s of
      Done ⇒ LNil
      | Skip·s' ⇒ unfold·h·s'
      | Yield·x·s' ⇒ LCons·x·(unfold·h·s')))

fixrec
  unfoldF :: ('a, 's) Stepper → ('s → 'a LList) → ('s → 'a LList)
where
  unfoldF·h·u·⊥ = ⊥
  | s ≠ ⊥ =>
    unfoldF·h·u·s =
    (case h·s of
      Done ⇒ LNil
      | Skip·s' ⇒ u·s'
      | Yield·x·s' ⇒ LCons·x·(u·s')))

lemma unfold-eq-fix: unfold·h = fix·(unfoldF·h)
proof (rule below-antisym)
  show unfold·h ⊑ fix·(unfoldF·h)
    apply (rule unfold.induct, simp, simp)
    apply (subst fix-eq)
    apply (rule cfun-belowI, rename-tac s)
    apply (case-tac s = ⊥, simp, simp)
    apply (intro monofun-cfun monofun-LAM below-refl, simp-all)
    done
  show fix·(unfoldF·h) ⊑ unfold·h
    apply (rule fix-ind, simp, simp)
    apply (subst unfold.unfold)
    apply (rule cfun-belowI, rename-tac s)
    apply (case-tac s = ⊥, simp, simp)
    apply (intro monofun-cfun monofun-LAM below-refl, simp-all)
    done
```

qed

```
lemma unfold-ind:
  fixes P :: ('s → 'a LList) ⇒ bool
  assumes adm P and P ⊥ and ⋀ u. P u ⇒ P (unfoldF · h · u)
  shows P (unfold · h)
unfolding unfold-eq-fix by (rule fix-ind [of P, OF assms])
```

fixrec

```
  unfold2 :: ('s → 'a LList) → ('a, 's) Step → 'a LList
```

where

```
  unfold2 · u · Done = LNil
  | s ≠ ⊥ ⇒ unfold2 · u · (Skip · s) = u · s
  | s ≠ ⊥ ⇒ unfold2 · u · (Yield · x · s) = LCons · x · (u · s)
```

```
lemma unfold2-strict [simp]: unfold2 · u · ⊥ = ⊥
by fixrec-simp
```

```
lemma unfold: s ≠ ⊥ ⇒ unfold · h · s = unfold2 · (unfold · h) · (h · s)
by (case-tac h · s, simp-all)
```

```
lemma unfoldF: s ≠ ⊥ ⇒ unfoldF · h · u · s = unfold2 · u · (h · s)
by (case-tac h · s, simp-all)
```

```
declare unfold.simps(2) [simp del]
declare unfoldF.simps(2) [simp del]
declare unfoldF [simp]
```

fixrec

```
  unstream :: ('a, 's) Stream → 'a LList
```

where

```
  s ≠ ⊥ ⇒ unstream · (Stream · h · s) = unfold · h · s
```

```
lemma unstream-strict [simp]: unstream · ⊥ = ⊥
by fixrec-simp
```

2.3 Converting from lists to streams

fixrec

```
  streamStep :: ('a LList) ⊥ → ('a, ('a LList) ⊥) Step
```

where

```
  streamStep · (up · LNil) = Done
  | streamStep · (up · (LCons · x · xs)) = Yield · x · (up · xs)
```

```
lemma streamStep-strict [simp]: streamStep · (up · ⊥) = ⊥
by fixrec-simp
```

fixrec

```
  stream :: 'a LList → ('a, ('a LList) ⊥) Stream
```

```

where
  stream·xs = Stream·streamStep·(up·xs)

lemma stream-defined [simp]: stream·xs ≠ ⊥
  by simp

lemma unstream-stream [simp]:
  fixes xs :: 'a LList
  shows unstream·(stream·xs) = xs
  by (induct xs, simp-all add: unfold)

declare stream.simps [simp del]

```

2.4 Bisimilarity relation on streams

```

definition
  bisimilar :: ('a, 's) Stream ⇒ ('a, 't) Stream ⇒ bool (infix ≈ 50)
where
  a ≈ b ⟷ unstream·a = unstream·b ∧ a ≠ ⊥ ∧ b ≠ ⊥

lemma unstream-cong:
  a ≈ b ⟹ unstream·a = unstream·b
  unfolding bisimilar-def by simp

lemma stream-cong:
  xs = ys ⟹ stream·xs ≈ stream·ys
  unfolding bisimilar-def by simp

lemma stream-unstream-cong:
  a ≈ b ⟹ stream·(unstream·a) ≈ b
  unfolding bisimilar-def by simp

end

```

3 Stream Fusion

```

theory StreamFusion
imports Stream
begin

```

```

3.1 Type constructors for state types

domain Switch = S1 | S2

domain 'a Maybe = Nothing | Just 'a

hide-const (open) Left Right

domain ('a, 'b) Either = Left 'a | Right 'b

```

```

domain ('a, 'b) Both (infixl ::!:> 25) = Both 'a 'b (infixl ::!:> 75)

domain 'a L = L (lazy 'a)

```

3.2 Map function

```

fixrec
  mapStep :: ('a → 'b) → ('s → ('a, 's) Step) → 's → ('b, 's) Step
where
  mapStep.f.h.⊥ = ⊥
  | s ≠ ⊥ ⇒ mapStep.f.h.s = (case h·s of
    Done ⇒ Done
    | Skip·s' ⇒ Skip·s'
    | Yield·x·s' ⇒ Yield·(f·x)·s')
fixrec
  mapS :: ('a → 'b) → ('a, 's) Stream → ('b, 's) Stream
where
  s ≠ ⊥ ⇒ mapS.f.(Stream·h·s) = Stream·(mapStep.f.h)·s

lemma unfold-mapStep:
  fixes f :: 'a → 'b and h :: 's → ('a, 's) Step
  assumes s ≠ ⊥
  shows unfold·(mapStep.f.h)·s = mapL.f·(unfold·h·s)
  proof (rule below-antisym)
    show unfold·(mapStep.f.h)·s ⊑ mapL.f·(unfold·h·s)
    using ⟨s ≠ ⊥⟩
    apply (induct arbitrary: s rule: unfold-ind [where h=mapStep.f.h])
    apply (simp, simp)
    apply (case-tac h·s, simp-all add: unfold)
    done
  next
    show mapL.f·(unfold·h·s) ⊑ unfold·(mapStep.f.h)·s
    using ⟨s ≠ ⊥⟩
    apply (induct arbitrary: s rule: unfold-ind [where h=h])
    apply (simp, simp)
    apply (case-tac h·s, simp-all add: unfold)
    done
  qed

lemma unstream-mapS:
  fixes f :: 'a → 'b and a :: ('a, 's) Stream
  shows a ≠ ⊥ ⇒ unstream·(mapS.f·a) = mapL.f·(unstream·a)
  by (induct a, simp, simp add: unfold-mapStep)

lemma mapS-defined: a ≠ ⊥ ⇒ mapS.f·a ≠ ⊥
  by (induct a, simp-all)

```

```

lemma mapS-cong:
  fixes f :: 'a → 'b
  fixes a :: ('a, 's) Stream
  fixes b :: ('a, 't) Stream
  shows f = g ⇒ a ≈ b ⇒ mapS·f·a ≈ mapS·g·b
unfolding bisimilar-def
by (simp add: unstream-mapS mapS-defined)

lemma mapL-eq: mapL·f·xs = unstream·(mapS·f·(stream·xs))
by (simp add: unstream-mapS)

```

3.3 Filter function

```

fixrec
  filterStep :: ('a → tr) → ('s → ('a, 's) Step) → 's → ('a, 's) Step
where
  filterStep·p·h·⊥ = ⊥
  | s ≠ ⊥ ⇒ filterStep·p·h·s = (case h·s of
    Done ⇒ Done
    | Skip·s' ⇒ Skip·s'
    | Yield·x·s' ⇒ (If p·x then Yield·x·s' else Skip·s'))
fixrec
  filterS :: ('a → tr) → ('a, 's) Stream → ('a, 's) Stream
where
  s ≠ ⊥ ⇒ filterS·p·(Stream·h·s) = Stream·(filterStep·p·h)·s

lemma unfold-filterStep:
  fixes p :: 'a → tr and h :: 's → ('a, 's) Step
  assumes s ≠ ⊥
  shows unfold·(filterStep·p·h)·s = filterL·p·(unfold·h·s)
proof (rule below-antisym)
  show unfold·(filterStep·p·h)·s ⊑ filterL·p·(unfold·h·s)
  using ‹s ≠ ⊥›
  apply (induct arbitrary: s rule: unfold-ind [where h=filterStep·p·h])
  apply (simp, simp)
  apply (case-tac h·s, simp-all add: unfold)
  apply (case-tac p·a rule: trE, simp-all)
  done
next
  show filterL·p·(unfold·h·s) ⊑ unfold·(filterStep·p·h)·s
  using ‹s ≠ ⊥›
  apply (induct arbitrary: s rule: unfold-ind [where h=h])
  apply (simp, simp)
  apply (case-tac h·s, simp-all add: unfold)
  apply (case-tac p·a rule: trE, simp-all add: unfold)
  done
qed

```

```

lemma unstream-filterS:
   $a \neq \perp \implies \text{unstream} \cdot (\text{filterS} \cdot p \cdot a) = \text{filterL} \cdot p \cdot (\text{unstream} \cdot a)$ 
by (induct a, simp, simp add: unfold-filterStep)
lemma filterS-defined:  $a \neq \perp \implies \text{filterS} \cdot p \cdot a \neq \perp$ 
by (induct a, simp-all)
lemma filterS-cong:
  fixes  $p :: 'a \rightarrow \text{tr}$ 
  fixes  $a :: ('a, 's) \text{ Stream}$ 
  fixes  $b :: ('a, 't) \text{ Stream}$ 
  shows  $p = q \implies a \approx b \implies \text{filterS} \cdot p \cdot a \approx \text{filterS} \cdot q \cdot b$ 
unfolding bisimilar-def
by (simp add: unstream-filterS filterS-defined)
lemma filterL-eq:  $\text{filterL} \cdot p \cdot xs = \text{unstream} \cdot (\text{filterS} \cdot p \cdot (\text{stream} \cdot xs))$ 
by (simp add: unstream-filterS)

```

3.4 Foldr function

```

fixrec
   $\text{foldrS} :: ('a \rightarrow 'b \rightarrow 'b) \rightarrow 'b \rightarrow ('a, 's) \text{ Stream} \rightarrow 'b$ 
where
  foldrS-Stream:
     $s \neq \perp \implies \text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s) =$ 
    (case h · s of Done  $\Rightarrow z$ 
     | Skip ·  $s' \Rightarrow \text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s')$ 
     | Yield ·  $x \cdot s' \Rightarrow f \cdot x \cdot (\text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s'))$ )
lemma unfold-foldrS:
  assumes  $s \neq \perp$  shows  $\text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s) = \text{foldrL} \cdot f \cdot z \cdot (\text{unfold} \cdot h \cdot s)$ 
proof (rule below-antisym)
  show  $\text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s) \sqsubseteq \text{foldrL} \cdot f \cdot z \cdot (\text{unfold} \cdot h \cdot s)$ 
  using  $\langle s \neq \perp \rangle$ 
  apply (induct arbitrary: s rule: foldrS.induct)
  apply (simp, simp, simp)
  apply (case-tac h · s, simp-all add: monofun-cfun unfold)
  done
next
  show  $\text{foldrL} \cdot f \cdot z \cdot (\text{unfold} \cdot h \cdot s) \sqsubseteq \text{foldrS} \cdot f \cdot z \cdot (\text{Stream} \cdot h \cdot s)$ 
  using  $\langle s \neq \perp \rangle$ 
  apply (induct arbitrary: s rule: unfold-ind)
  apply (simp, simp)
  apply (case-tac h · s, simp-all add: monofun-cfun unfold)
  done
qed
lemma unstream-foldrS:
   $a \neq \perp \implies \text{foldrS} \cdot f \cdot z \cdot a = \text{foldrL} \cdot f \cdot z \cdot (\text{unstream} \cdot a)$ 

```

```
by (induct a, simp, simp del: foldrS-Stream add: unfold-foldrS)
```

```
lemma foldrS-cong:
  fixes a :: ('a, 's) Stream
  fixes b :: ('a, 't) Stream
  shows f = g  $\Rightarrow$  z = w  $\Rightarrow$  a  $\approx$  b  $\Rightarrow$  foldrS.f.z.a = foldrS.g.w.b
by (simp add: bisimilar-def unstream-foldrS)
```

```
lemma foldrL-eq:
  foldrL.f.z.xs = foldrS.f.z.(stream.xs)
by (simp add: unstream-foldrS)
```

3.5 EnumFromTo function

```
type-synonym int' = int⊥
```

```
fixrec
  enumFromToStep :: int'  $\rightarrow$  (int')⊥  $\rightarrow$  (int', (int')⊥) Step
where
  enumFromToStep.(up.y).(up.(up.x)) =
    (if x  $\leq$  y then Yield.(up.x).(up.(up.(x+1))) else Done)
```

```
lemma enumFromToStep-strict [simp]:
  enumFromToStep. $\perp$ .x'' =  $\perp$ 
  enumFromToStep.(up.y). $\perp$  =  $\perp$ 
  enumFromToStep.(up.y).(up. $\perp$ ) =  $\perp$ 
by fixrec-simp+
```

lemma enumFromToStep-simps' [simp]:
 $x \leq y \Rightarrow \text{enumFromToStep}.\text{(up.y)}.\text{(up.(up.x))} =$
 $\text{Yield}.\text{(up.x)}.\text{(up.(up.(x+1)))}$
 $\neg x \leq y \Rightarrow \text{enumFromToStep}.\text{(up.y)}.\text{(up.(up.x))} = \text{Done}$
by simp-all

```
declare enumFromToStep.simps [simp del]
```

```
fixrec
  enumFromToS :: int'  $\rightarrow$  int'  $\rightarrow$  (int', (int')⊥) Stream
where
  enumFromToS.x.y = Stream.(enumFromToStep.y).(up.x)
```

```
declare enumFromToS.simps [simp del]
```

```
lemma unfold-enumFromToStep:
  unfold.(enumFromToStep.(up.y)).(up.n) = enumFromToL.n.(up.y)
proof (rule below-antisym)
  show unfold.(enumFromToStep.(up.y)).(up.n)  $\sqsubseteq$  enumFromToL.n.(up.y)
  apply (induct arbitrary: n rule: unfold-ind [where h=enumFromToStep.(up.y)])
  apply (simp, simp)
```

```

apply (case-tac n, simp, simp)
apply (case-tac x ≤ y, simp-all)
done
next
show enumFromToL·n·(up·y) ⊑ unfold·(enumFromToStep·(up·y))·(up·n)
apply (induct arbitrary: n rule: enumFromToL.induct)
apply (simp, simp)
apply (rename-tac e n)
apply (case-tac n, simp)
apply (case-tac x ≤ y, simp-all add: unfold)
done
qed

lemma unstream-enumFromToS:
unstream·(enumFromToS·x·y) = enumFromToL·x·y
apply (simp add: enumFromToS.simps)
apply (induct y, simp add: unfold)
apply (induct x, simp add: unfold)
apply (simp add: unfold-enumFromToStep)
done

lemma enumFromToS-defined: enumFromToS·x·y ≠ ⊥
by (simp add: enumFromToS.simps)

lemma enumFromToS-cong:
x = x' ⇒ y = y' ⇒ enumFromToS·x·y ≈ enumFromToS·x'·y'
unfolding bisimilar-def by (simp add: enumFromToS-defined)

lemma enumFromToL-eq: enumFromToL·x·y = unstream·(enumFromToS·x·y)
by (simp add: unstream-enumFromToS)

```

3.6 Append function

```

fixrec
appendStep :: 
('s → ('a, 's) Step) →
('t → ('a, 't) Step) →
't → ('s, 't) Either → ('a, ('s, 't) Either) Step
where
sa ≠ ⊥ ⇒ appendStep·ha·hb·sb0·(Left·sa) =
(case ha·sa of
  Done ⇒ Skip·(Right·sb0)
  | Skip·sa' ⇒ Skip·(Left·sa')
  | Yield·x·sa' ⇒ Yield·x·(Left·sa'))
| sb ≠ ⊥ ⇒ appendStep·ha·hb·sb0·(Right·sb) =
(case hb·sb of
  Done ⇒ Done
  | Skip·sb' ⇒ Skip·(Right·sb')
  | Yield·x·sb' ⇒ Yield·x·(Right·sb'))

```

```

lemma appendStep-strict [simp]: appendStep·ha·hb·sb0· $\perp$  =  $\perp$ 
by fixrec-simp

fixrec
  appendS :: ('a, 's) Stream → ('a, 't) Stream → ('a, ('s, 't) Either) Stream
where
  sa0 ≠  $\perp$   $\implies$  sb0 ≠  $\perp$   $\implies$ 
    appendS·(Stream·ha·sa0)·(Stream·hb·sb0) =
      Stream·(appendStep·ha·hb·sb0)·(Left·sa0)

lemma unfold-appendStep:
  fixes ha :: 's → ('a, 's) Step
  fixes hb :: 't → ('a, 't) Step
  assumes sb0 [simp]: sb0 ≠  $\perp$ 
  shows
    ( $\forall$  sa. sa ≠  $\perp$   $\longrightarrow$  unfold·(appendStep·ha·hb·sb0)·(Left·sa) =
      appendL·(unfold·ha·sa)·(unfold·hb·sb0))  $\wedge$ 
    ( $\forall$  sb. sb ≠  $\perp$   $\longrightarrow$  unfold·(appendStep·ha·hb·sb0)·(Right·sb) =
      unfold·hb·sb)
proof –
  note unfold [simp]
  let ?h = appendStep·ha·hb·sb0

  have 1:
  ( $\forall$  sa. sa ≠  $\perp$   $\longrightarrow$ 
    unfold·?h·(Left·sa)  $\sqsubseteq$ 
    appendL·(unfold·ha·sa)·(unfold·hb·sb0))
   $\wedge$ 
  ( $\forall$  sb. sb ≠  $\perp$   $\longrightarrow$  unfold·?h·(Right·sb)  $\sqsubseteq$  unfold·hb·sb)
  apply (rule unfold-ind [where h=?h])
  apply simp
  apply simp
  apply (intro conjI allI impI)
  apply (case-tac ha·sa, simp, simp, simp, simp)
  apply (case-tac hb·sb, simp, simp, simp, simp)
  done

  let ?P =  $\lambda$ ua ub.  $\forall$  sa. sa ≠  $\perp$   $\longrightarrow$ 
    appendL·(ua·sa)·(ub·sb0)  $\sqsubseteq$  unfold·?h·(Left·sa)

  let ?Q =  $\lambda$ ub.  $\forall$  sb. sb ≠  $\perp$   $\longrightarrow$  ub·sb  $\sqsubseteq$  unfold·?h·(Right·sb)

  have P-base:  $\bigwedge$ ub. ?P ⊥ ub
  by simp

  have Q-base: ?Q ⊥
  by simp

```

```

have P-step:  $\bigwedge ua\ ub.\ ?P\ ua\ ub \implies ?Q\ ub \implies ?P\ (\text{unfoldF}\cdot ha\cdot ua)\ ub$ 
  apply (intro allI impI)
  apply (case-tac ha·sa, simp, simp, simp, simp)
  done

have Q-step:  $\bigwedge ua\ ub.\ ?Q\ ub \implies ?Q\ (\text{unfoldF}\cdot hb\cdot ub)$ 
  apply (intro allI impI)
  apply (case-tac hb·sb, simp, simp, simp, simp)
  done

have Q:  $?Q\ (\text{unfold}\cdot hb)$ 
  apply (rule unfold-ind [where h=hb], simp)
  apply (rule Q-base)
  apply (erule Q-step)
  done

have P:  $?P\ (\text{unfold}\cdot ha) (\text{unfold}\cdot hb)$ 
  apply (rule unfold-ind [where h=ha], simp)
  apply (rule P-base)
  apply (erule P-step)
  apply (rule Q)
  done

have 2:  $?P\ (\text{unfold}\cdot ha) (\text{unfold}\cdot hb) \wedge ?Q\ (\text{unfold}\cdot hb)$ 
  using P Q by (rule conjI)

from 1 2 show ?thesis
  by (simp add: po-eq-conv [where 'a='a LList])
qed

lemma appendS-defined:  $xs \neq \perp \implies ys \neq \perp \implies \text{appendS}\cdot xs\cdot ys \neq \perp$ 
by (cases xs, simp, cases ys, simp, simp)

lemma unstream-appendS:
 $a \neq \perp \implies b \neq \perp \implies$ 
 $\text{unstream}\cdot(\text{appendS}\cdot a\cdot b) = \text{appendL}\cdot(\text{unstream}\cdot a)\cdot(\text{unstream}\cdot b)$ 
apply (cases a, simp, cases b, simp)
apply (simp add: unfold-appendStep)
done

lemma appendS-cong:
fixes f :: 'a → 'b
fixes a :: ('a, 's) Stream
fixes b :: ('a, 't) Stream
shows a ≈ a' ⇒ b ≈ b' ⇒ appendS·a·b ≈ appendS·a'·b'
unfolding bisimilar-def
by (simp add: unstream-appendS appendS-defined)

```

lemma *appendL-eq*: $\text{appendL} \cdot xs \cdot ys = \text{unstream} \cdot (\text{appendS} \cdot (\text{stream} \cdot xs) \cdot (\text{stream} \cdot ys))$
by (*simp add: unstream-appendS*)

3.7 ZipWith function

fixrec
 $\text{zipWithStep} ::$
 $('a \rightarrow 'b \rightarrow 'c) \rightarrow$
 $('s \rightarrow ('a, 's) \text{ Step}) \rightarrow$
 $('t \rightarrow ('b, 't) \text{ Step}) \rightarrow$
 $'s :: 't :: 'a \text{ L Maybe} \rightarrow ('c, 's :: 't :: 'a \text{ L Maybe}) \text{ Step}$

where

$sa \neq \perp \Rightarrow sb \neq \perp \Rightarrow$
 $\text{zipWithStep}.f \cdot ha \cdot hb \cdot (sa :: sb :: \text{Nothing}) =$
 $(\text{case } ha \cdot sa \text{ of}$
 $\quad \text{Done} \Rightarrow \text{Done}$
 $\quad | \text{Skip} \cdot sa' \Rightarrow \text{Skip} \cdot (sa' :: sb :: \text{Nothing})$
 $\quad | \text{Yield} \cdot a \cdot sa' \Rightarrow \text{Skip} \cdot (sa' :: sb :: \text{Just} \cdot (L \cdot a))$
 $| sa \neq \perp \Rightarrow sb \neq \perp \Rightarrow$
 $\quad \text{zipWithStep}.f \cdot ha \cdot hb \cdot (sa :: sb :: \text{Just} \cdot (L \cdot a)) =$
 $\quad (\text{case } hb \cdot sb \text{ of}$
 $\quad \quad \text{Done} \Rightarrow \text{Done}$
 $\quad \quad | \text{Skip} \cdot sb' \Rightarrow \text{Skip} \cdot (sa :: sb' :: \text{Just} \cdot (L \cdot a))$
 $\quad \quad | \text{Yield} \cdot b \cdot sb' \Rightarrow \text{Yield} \cdot (f \cdot a \cdot b) \cdot (sa :: sb' :: \text{Nothing}))$

lemma *zipWithStep-strict* [*simp*]: $\text{zipWithStep}.f \cdot ha \cdot hb \cdot \perp = \perp$
by *fixrec-simp*

fixrec
 $\text{zipWithS} :: ('a \rightarrow 'b \rightarrow 'c) \rightarrow$
 $('a, 's) \text{ Stream} \rightarrow ('b, 't) \text{ Stream} \rightarrow ('c, 's :: 't :: 'a \text{ L Maybe}) \text{ Stream}$
where
 $sa0 \neq \perp \Rightarrow sb0 \neq \perp \Rightarrow \text{zipWithS}.f \cdot (\text{Stream} \cdot ha \cdot sa0) \cdot (\text{Stream} \cdot hb \cdot sb0) =$
 $\text{Stream} \cdot (\text{zipWithStep}.f \cdot ha \cdot hb) \cdot (sa0 :: sb0 :: \text{Nothing})$

lemma *zipWithS-fix-ind-lemma*:
fixes $P Q :: \text{nat} \Rightarrow \text{nat} \Rightarrow \text{bool}$
assumes $P\text{-}0: \bigwedge j. P\ 0\ j \text{ and } P\text{-Suc}: \bigwedge i\ j. P\ i\ j \Rightarrow Q\ i\ j \Rightarrow P\ (\text{Suc}\ i)\ j$
assumes $Q\text{-}0: \bigwedge i. Q\ i\ 0 \text{ and } Q\text{-Suc}: \bigwedge i\ j. P\ i\ j \Rightarrow Q\ i\ j \Rightarrow Q\ i\ (\text{Suc}\ j)$
shows $P\ i\ j \wedge Q\ i\ j$
apply (*induct n ≡ i + j arbitrary: i j*)
apply (*simp add: P-0 Q-0*)
apply (*rule conjI*)
apply (*case-tac i, simp add: P-0, simp add: P-Suc*)
apply (*case-tac j, simp add: Q-0, simp add: Q-Suc*)
done

lemma *zipWithS-fix-ind*:
assumes $x: x = \text{fix} \cdot f \text{ and } y: y = \text{fix} \cdot g$

```

assumes adm-P: adm ( $\lambda x. P (\text{fst } x) (\text{snd } x)$ )
assumes adm-Q: adm ( $\lambda x. Q (\text{fst } x) (\text{snd } x)$ )
assumes P-0:  $\bigwedge b. P \perp b$  and P-Suc:  $\bigwedge a b. P a b \implies Q a b \implies P (f \cdot a) b$ 
assumes Q-0:  $\bigwedge a. Q a \perp$  and Q-Suc:  $\bigwedge a b. P a b \implies Q a b \implies Q a (g \cdot b)$ 
shows P x y  $\wedge$  Q x y
proof -
have 1:  $\bigwedge i j. P (\text{iterate } i \cdot f \cdot \perp) (\text{iterate } j \cdot g \cdot \perp) \wedge Q (\text{iterate } i \cdot f \cdot \perp) (\text{iterate } j \cdot g \cdot \perp)$ 
apply (rule-tac i=i and j=j in zipWithS-fix-ind-lemma)
apply (simp add: P-0)
apply (simp add: P-Suc)
apply (simp add: Q-0)
apply (simp add: Q-Suc)
done
have case-prod P ( $\bigsqcup i. (\text{iterate } i \cdot f \cdot \perp, \text{iterate } i \cdot g \cdot \perp)$ )
apply (rule admD)
apply (simp add: split-def adm-P)
apply simp
apply (simp add: 1)
done
then have P: P x y
unfolding x y fix-def2
by (simp add: lub-prod)
have case-prod Q ( $\bigsqcup i. (\text{iterate } i \cdot f \cdot \perp, \text{iterate } i \cdot g \cdot \perp)$ )
apply (rule admD)
apply (simp add: split-def adm-Q)
apply simp
apply (simp add: 1)
done
then have Q: Q x y
unfolding x y fix-def2
by (simp add: lub-prod)
from P Q show ?thesis by simp
qed

```

```

lemma unfold-zipWithStep:
fixes f :: 'a  $\rightarrow$  'b  $\rightarrow$  'c
fixes ha :: 's  $\rightarrow$  ('a, 's) Step
fixes hb :: 't  $\rightarrow$  ('b, 't) Step
defines h-def: h  $\equiv$  zipWithStep f ha hb
shows
 $(\forall sa sb. sa \neq \perp \longrightarrow sb \neq \perp \longrightarrow$ 
 $\text{unfold}\cdot h\cdot (sa :: sb :: \text{Nothing}) =$ 
 $\text{zipWithL}\cdot f\cdot (\text{unfold}\cdot ha\cdot sa)\cdot (\text{unfold}\cdot hb\cdot sb)) \wedge$ 
 $(\forall sa sb a. sa \neq \perp \longrightarrow sb \neq \perp \longrightarrow$ 
 $\text{unfold}\cdot h\cdot (sa :: sb :: \text{Just}\cdot (L\cdot a)) =$ 
 $\text{zipWithL}\cdot f\cdot (L\text{Cons}\cdot a\cdot (\text{unfold}\cdot ha\cdot sa))\cdot (\text{unfold}\cdot hb\cdot sb))$ 
proof -
note unfold [simp]
have h-simps [simp]:

```

```

 $\wedge_{sa\ sb. sa \neq \perp} \Rightarrow sb \neq \perp \Rightarrow h\cdot(sa :: sb :: Nothing) =$ 
 $(\text{case } ha\cdot sa \text{ of}$ 
 $\quad Done \Rightarrow Done$ 
 $\quad | Skip\cdot sa' \Rightarrow Skip\cdot(sa' :: sb :: Nothing)$ 
 $\quad | Yield\cdot a\cdot sa' \Rightarrow Skip\cdot(sa' :: sb :: Just\cdot(L\cdot a))$ 
 $\wedge_{sa\ sb\ a. sa \neq \perp} \Rightarrow sb \neq \perp \Rightarrow h\cdot(sa :: sb :: Just\cdot(L\cdot a)) =$ 
 $(\text{case } hb\cdot sb \text{ of}$ 
 $\quad Done \Rightarrow Done$ 
 $\quad | Skip\cdot sb' \Rightarrow Skip\cdot(sa :: sb' :: Just\cdot(L\cdot a))$ 
 $\quad | Yield\cdot b\cdot sb' \Rightarrow Yield\cdot(f\cdot a\cdot b)\cdot(sa :: sb' :: Nothing))$ 
 $h\cdot \perp = \perp$ 
unfolding  $h\text{-def}$  by simp-all

```

have 1:

```

 $(\forall sa\ sb. sa \neq \perp \rightarrow sb \neq \perp \rightarrow$ 
 $\quad unfold\cdot h\cdot(sa :: sb :: Nothing) \sqsubseteq$ 
 $\quad zipWithL\cdot f\cdot(unfold\cdot ha\cdot sa)\cdot(unfold\cdot hb\cdot sb))$ 
 $\wedge$ 
 $(\forall sa\ sb\ a. sa \neq \perp \rightarrow sb \neq \perp \rightarrow$ 
 $\quad unfold\cdot h\cdot(sa :: sb :: Just\cdot(L\cdot a)) \sqsubseteq$ 
 $\quad zipWithL\cdot f\cdot(LCons\cdot a\cdot(unfold\cdot ha\cdot sa))\cdot(unfold\cdot hb\cdot sb))$ 
apply (rule unfold-ind [where  $h=h$ ], simp)
apply simp
apply (intro conjI allI impI)
apply (case-tac  $ha\cdot sa$ , simp, simp, simp, simp)
apply (case-tac  $hb\cdot sb$ , simp, simp, simp, simp)
done

```

```

let  $?P = \lambda ua\ ub. \forall sa\ sb. sa \neq \perp \rightarrow sb \neq \perp \rightarrow$ 
 $\quad zipWithL\cdot f\cdot(ua\cdot sa)\cdot(ub\cdot sb) \sqsubseteq unfold\cdot h\cdot(sa :: sb :: Nothing)$ 

```

```

let  $?Q = \lambda ua\ ub. \forall sa\ sb\ a. sa \neq \perp \rightarrow sb \neq \perp \rightarrow$ 
 $\quad zipWithL\cdot f\cdot(LCons\cdot a\cdot(ua\cdot sa))\cdot(ub\cdot sb) \sqsubseteq$ 
 $\quad unfold\cdot h\cdot(sa :: sb :: Just\cdot(L\cdot a))$ 

```

have P-base: $\bigwedge_{ub. ?P \perp ub}$
by *simp*

have Q-base: $\bigwedge_{ua. ?Q ua \perp}$
by *simp*

have P-step: $\bigwedge_{ua\ ub. ?P ua\ ub} \Rightarrow ?Q ua\ ub \Rightarrow ?P (unfoldF\cdot ha\cdot ua) \perp ub$
by (*clarsimp*, *case-tac* $ha\cdot sa$, *simp-all*)

have Q-step: $\bigwedge_{ua\ ub. ?P ua\ ub} \Rightarrow ?Q ua\ ub \Rightarrow ?Q ua (unfoldF\cdot hb\cdot ub)$
by (*clarsimp*, *case-tac* $hb\cdot sb$, *simp-all*)

have 2: $?P (unfold\cdot ha) (unfold\cdot hb) \wedge ?Q (unfold\cdot ha) (unfold\cdot hb)$
apply (*rule zipWithS-fix-ind* [*OF* *unfold-eq-fix* [*of ha*] *unfold-eq-fix* [*of hb*]])

```

apply (simp, simp)
apply (rule P-base)
apply (erule (1) P-step)
apply (rule Q-base)
apply (erule (1) Q-step)
done

from 1 2 show ?thesis
  by (simp-all add: po-eq-conv [where 'a='c LList])
qed

lemma zipWithS-defined:  $a \neq \perp \Rightarrow b \neq \perp \Rightarrow \text{zipWithS}\cdot f\cdot a\cdot b \neq \perp$ 
  by (cases a, simp, cases b, simp, simp)

lemma unstream-zipWithS:
   $a \neq \perp \Rightarrow b \neq \perp \Rightarrow$ 
     $\text{unstream}\cdot(\text{zipWithS}\cdot f\cdot a\cdot b) = \text{zipWithL}\cdot f\cdot(\text{unstream}\cdot a)\cdot(\text{unstream}\cdot b)$ 
  apply (cases a, simp, cases b, simp)
  apply (simp add: unfold-zipWithStep)
  done

lemma zipWithS-cong:
   $f = f' \Rightarrow a \approx a' \Rightarrow b \approx b' \Rightarrow$ 
     $\text{zipWithS}\cdot f\cdot a\cdot b \approx \text{zipWithS}\cdot f\cdot a'\cdot b'$ 
  unfolding bisimilar-def
  by (simp add: unstream-zipWithS zipWithS-defined)

lemma zipWithL Eq:
   $\text{zipWithL}\cdot f\cdot xs\cdot ys = \text{unstream}\cdot(\text{zipWithS}\cdot f\cdot(\text{stream}\cdot xs)\cdot(\text{stream}\cdot ys))$ 
  by (simp add: unstream-zipWithS)

```

3.8 ConcatMap function

```

fixrec
  concatMapStep :: 
    ('a → ('b, 't) Stream) →
    ('s → ('a, 's) Step) →
    's :: ('b, 't) Stream Maybe →
    ('b, 's :: ('b, 't) Stream Maybe) Step
  where
     $sa \neq \perp \Rightarrow \text{concatMapStep}\cdot f\cdot ha\cdot(sa :: Nothing) =$ 
      (case ha·sa of
        Done ⇒ Done
         $| \text{Skip}\cdot sa' \Rightarrow \text{Skip}\cdot(sa' :: Nothing)$ 
         $| \text{Yield}\cdot a\cdot sa' \Rightarrow \text{Skip}\cdot(sa' :: Just\cdot(f\cdot a))$ 
      )
     $| sa \neq \perp \Rightarrow sb \neq \perp \Rightarrow$ 
       $\text{concatMapStep}\cdot f\cdot ha\cdot(sa :: Just\cdot(Stream\cdot hb\cdot sb)) =$ 
      (case hb·sb of
        Done ⇒ Skip·(sa :: Nothing)
      )

```

```

| Skip·sb' ⇒ Skip·(sa :: Just·(Stream·hb·sb'))
| Yield·b·sb' ⇒ Yield·b·(sa :: Just·(Stream·hb·sb')))
```

lemma concatMapStep-strict [simp]: concatMapStep·f·ha· \perp = \perp
by fixrec-simp

fixrec

```

concatMapS :: ('a → ('b, 't) Stream) → ('a, 's) Stream →
('b, 's :: ('b, 't) Stream Maybe) Stream
```

where

```

s ≠  $\perp$  ⇒ concatMapS·f·(Stream·h·s) = Stream·(concatMapStep·f·h)·(s :: Nothing)
```

lemma concatMapS-strict [simp]: concatMapS·f· \perp = \perp
by fixrec-simp

lemma unfold-concatMapStep:

fixes ha :: 's → ('a, 's) Step

fixes f :: 'a → ('b, 't) Stream

defines h-def: h ≡ concatMapStep·f·ha

defines f'-def: f' ≡ unstream oo f

shows

```

(∀ sa. sa ≠  $\perp$  →
unfold·h·(sa :: Nothing) = concatMapL·f'·(unfold·ha·sa)) ∧
(∀ sa hb sb. sa ≠  $\perp$  → sb ≠  $\perp$  →
unfold·h·(sa :: Just·(Stream·hb·sb)) =
appendL·(unfold·hb·sb)·(concatMapL·f'·(unfold·ha·sa)))
```

proof –

note unfold [simp]

have h-simps [simp]:

```

 $\wedge$  sa. sa ≠  $\perp$  ⇒ h·(sa :: Nothing) =
(case ha·sa of Done ⇒ Done
| Skip·sa' ⇒ Skip·(sa' :: Nothing)
| Yield·a·sa' ⇒ Skip·(sa' :: Just·(f·a)))
```

```

 $\wedge$  sa hb sb. sa ≠  $\perp$  ⇒ sb ≠  $\perp$  ⇒ h·(sa :: Just·(Stream·hb·sb)) =
(case hb·sb of Done ⇒ Skip·(sa :: Nothing)
| Skip·sb' ⇒ Skip·(sa :: Just·(Stream·hb·sb'))
| Yield·b·sb' ⇒ Yield·b·(sa :: Just·(Stream·hb·sb')))
```

h· \perp = \perp

unfolding h-def **by** simp-all

have f'-beta [simp]: \wedge a. f'·a = unstream·(f·a)

unfolding f'-def **by** simp

have 1:

```

(∀ sa. sa ≠  $\perp$  →
unfold·h·(sa :: Nothing) ⊑ concatMapL·f'·(unfold·ha·sa))
 $\wedge$ 
```

```


$$(\forall sa\ hb\ sb.\ sa \neq \perp \longrightarrow sb \neq \perp \longrightarrow$$


$$\quad unfold\cdot h\cdot(sa :: Just\cdot(Stream\cdot hb\cdot sb)) \sqsubseteq$$


$$\quad appendL\cdot(unfold\cdot hb\cdot sb)\cdot(concatMapL\cdot f'\cdot(unfold\cdot ha\cdot sa)))$$

apply (rule unfold-ind [where h=h], simp)
apply simp
apply (intro conjI allI impI)
apply (case-tac ha·sa, simp, simp, simp)
apply (rename-tac a sa')
apply (case-tac f·a, simp, simp)
apply (case-tac hb·sb, simp, simp, simp, simp)
done

let  $?P = \lambda ua.\ \forall sa.\ sa \neq \perp \longrightarrow$ 

$$\quad concatMapL\cdot f'\cdot(ua\cdot sa) \sqsubseteq unfold\cdot h\cdot(sa :: Nothing)$$


let  $?Q = \lambda hb\ ua\ ub.\ \forall sa\ sb.\ sa \neq \perp \longrightarrow sb \neq \perp \longrightarrow$ 

$$\quad appendL\cdot(ub\cdot sb)\cdot(concatMapL\cdot f'\cdot(ua\cdot sa)) \sqsubseteq$$


$$\quad unfold\cdot h\cdot(sa :: Just\cdot(Stream\cdot hb\cdot sb))$$


have P-base:  $?P \perp$ 
by simp

have P-step:  $\bigwedge ua.\ ?P\ ua \implies \forall hb.\ ?Q\ hb\ ua\ (unfold\cdot hb) \implies ?P\ (unfoldF\cdot ha\cdot ua)$ 
apply (intro allI impI)
apply (case-tac ha·sa, simp, simp, simp)
apply (rename-tac a sa')
apply (case-tac f·a, simp, simp)
done

have Q-base:  $\bigwedge ua\ hb.\ ?Q\ hb\ ua \perp$ 
by simp

have Q-step:  $\bigwedge hb\ ua\ ub.\ ?P\ ua \implies ?Q\ hb\ ua\ ub \implies ?Q\ hb\ ua\ (unfoldF\cdot hb\cdot ub)$ 
apply (intro allI impI)
apply (case-tac hb·sb, simp, simp, simp, simp)
done

have 2:  $?P\ (unfold\cdot ha) \wedge (\forall hb.\ ?Q\ hb\ (unfold\cdot ha)\ (unfold\cdot hb))$ 
apply (rule unfold-ind [where h=ha], simp)
apply (rule conjI)
apply (rule P-base)
apply (rule allI, rule-tac h=hb in unfold-ind, simp)
apply (rule Q-base)
apply (erule Q-step [OF P-base])
apply (erule conjE)
apply (rule conjI)
apply (erule (1) P-step)
apply (rule allI, rule-tac h=hb in unfold-ind, simp)
apply (rule Q-base)

```

```

apply (erule (2) Q-step [OF P-step])
done

from 1 2 show ?thesis
  by (simp-all add: po-eq-conv [where 'a='b LList])
qed

lemma unstream-concatMapS:
  unstream·(concatMapS·f·a) = concatMapL·(unstream oo f)·(unstream·a)
  by (cases a, simp, simp add: unfold-concatMapStep)

lemma concatMapS-defined: a ≠ ⊥  $\Rightarrow$  concatMapS·f·a ≠ ⊥
  by (induct a, simp-all)

lemma concatMapS-cong:
  fixes f :: 'a  $\Rightarrow$  ('b, 's) Stream
  fixes g :: 'a  $\Rightarrow$  ('b, 't) Stream
  fixes a :: ('a, 'u) Stream
  fixes b :: ('a, 'v) Stream
  shows ( $\bigwedge x. f x \approx g x$ )  $\Rightarrow$  a ≈ b  $\Rightarrow$  cont f  $\Rightarrow$  cont g  $\Rightarrow$ 
    concatMapS·( $\Lambda x. f x$ )·a ≈ concatMapS·( $\Lambda x. g x$ )·b
  unfolding bisimilar-def
  by (simp add: unstream-concatMapS oo-def concatMapS-defined)

lemma concatMapL-eq:
  concatMapL·f·xs = unstream·(concatMapS·(stream oo f)·(stream·xs))
  by (simp add: unstream-concatMapS oo-def eta-cfun)

```

3.9 Examples

```

lemmas stream-eqs =
  mapL-eq
  filterL-eq
  foldrL-eq
  enumFromToL-eq
  appendL-eq
  zipWithL-eq
  concatMapL-eq

lemmas stream-congs =
  unstream-cong
  stream-cong
  stream-unstream-cong
  mapS-cong
  filterS-cong
  foldrS-cong
  enumFromToS-cong
  appendS-cong
  zipWithS-cong

```

```

concatMapS-cong

lemma
  mapL·f oo filterL·p oo mapL·g =
  unstream oo mapS·f oo filterS·p oo mapS·g oo stream
apply (rule cfun-eqI, simp)
apply (unfold stream-eqs)
apply (intro stream-congs refl)
done

lemma
  foldrL·f·z·(mapL·g·(filterL·p·(enumFromToL·x·y))) =
  foldrS·f·z·(mapS·g·(filterS·p·(enumFromToS·x·y)))
apply (unfold stream-eqs)
apply (intro stream-congs refl)
done

lemma oo-LAM [simp]: cont g ==> f oo (Λ x. g x) = (Λ x. f·(g x))
  unfolding oo-def by simp

lemma
  concatMapL·(Λ k.
    mapL·(Λ m. f·k·m)·(enumFromToL·one·k))·(enumFromToL·one·n) =
  unstream·(concatMapS·(Λ k.
    mapS·(Λ m. f·k·m)·(enumFromToS·one·k))·(enumFromToS·one·n))
  unfolding stream-eqs
apply simp
apply (simp add: stream-congs)
done

end

```

References

- [1] D. Coutts, R. Leshchinskiy, and D. Stewart. Stream fusion: From lists to streams to nothing at all. In *Proceedings of the ACM SIGPLAN International Conference on Functional Programming, ICFP 2007*, Apr. 2007.